



Oct

19

# How to calculate age in Excel: from date of birth, between two dates, and more

by Svetlana Cheusheva | updated on September 24, 2020

86 Comments

*The tutorial shows different ways to get age from birthday in Excel. You will learn a handful of formulas to calculate age as a number of complete years, get exact age in years, months and days at today's date or a particular date.*

There is no special function to calculate age in Excel, however there exist a few different ways to convert date of birth to age. This tutorial will explain the advantages and drawbacks of each way, shows how to make a perfect age calculation formula in Excel and tweak it for solving some specific tasks.

## [How to calculate age from date of birth in Excel](#)

[Calculate age as the number of full years](#)[Get age from birth date in years, months, and days](#)

## [More examples of age formula in Excel](#)

[Calculate age on a specific date](#)[Get age in a certain year](#)[Compute when a person attains N years of age](#)[Get age from day, month and year in different cells](#)

## [Age calculator in Excel](#)

[How to create an age calculator in Excel](#)[Ready-to-use age calculator for Excel](#)[Highlight ages under or over a particular age](#)

## How to calculate age from date of birth in Excel

In everyday life, the question "How old are you?" usually implies an answer indicating how many years you have been alive. In Microsoft Excel, you can make a formula to compute an exact age in months, days, hours and even minutes. But let's be traditional, and learn how to calculate age from DOB in years first.

### Basic Excel formula for age in years

How do you normally figure out someone's age? Simply by subtracting the birth date from the current date. This conventional age formula can also be used in Excel.

Assuming a birthdate is in cell B2, the formula to calculate age in years goes as follows:

```
= (TODAY () -B2) /365
```

The first part of the formula (TODAY()-B2) returns the difference between the current date and date of birth in days, and then you divide that number by 365 to get the numbers of years.

The formula is obvious and easy-to-remember, however, there is a tiny problem. In most cases, it returns a decimal number as demonstrated in the screenshot below.

C2 : =(TODAY()-B2)/365

	A	B	C
1	Name	Date of birth	Age
2	Mike	6-Jun-1996	18.90685
3	Natali	20-Apr-1983	32.04658
4	Neal	16-Nov-2000	14.45753
5	Peter	21-Aug-1984	30.70685
6	Kate	30-Mar-1990	25.09863

To display the number of complete years, use the [INT function](#) to round a decimal down to the nearest integer:

```
=INT ( (TODAY () -B2) /365)
```

C2 : =INT((TODAY()-B2)/365)

	A	B	C	D
1	Name	Date of birth	Age	
2	Mike	6-Jun-1996	18	
3	Natali	20-Apr-1983	32	
4	Neal	16-Nov-2000	14	
5	Peter	21-Aug-1984	30	
6	Kate	30-Mar-1990	25	

**Drawbacks:** Using this age formula in Excel produces pretty accurate results, but not flawless. Dividing by the average number of days in a year works fine most of the time, but sometimes it gets the age wrong. For example, if someone was born on February 29 and today is February 28, the formula will make a person one day older.

As an alternative, you can divide by 365.25 instead of 365 since every fourth year has 366 days. However, this approach is not perfect either. For example, if you are calculating the age of a child who hasn't yet lived through a leap year, dividing by 365.25 produces a wrong result.

Overall, subtracting the birth date from the current date works great in normal life, but is not the ideal approach in Excel. Further on in this tutorial, you will learn a couple of special functions that calculate age faultlessly regardless of the year.

## Calculate age from date of birth with YEARFRAC function

A more reliable way to convert DOB to age in Excel is using the YEARFRAC function that returns the fraction of the year, i.e. the number of whole days between two dates.

The syntax of the YEARFRAC function is as follows:

```
YEARFRAC(start_date, end_date, [basis])
```

The first two arguments are obvious and hardly require any additional explanation. *Basis* is an optional argument that defines the day count basis to use.

To make a perfectly true age formula, supply the following values to the YEARFRAC function:

**Start\_date** - date of birth.

**End\_date** - TODAY() function to return the today's date.

**Basis** - use basis 1 that tells Excel to divide the actual number of days per month by the actual number of days per year.

Considering the above, an Excel formula to calculate age from date of birth is as follows:

```
YEARFRAC(date of birth, TODAY(), 1)
```

Assuming the birthdate is in cell B2, the formula takes the following shape:

```
=YEARFRAC(B2, TODAY(), 1)
```

	A	B	C
1	Name	Date of birth	Age
2	Mike	6-Jun-1996	20.29644
3	Natalie	20-Mar-1983	33.51268
4	Neal	16-Nov-2000	15.85024
5	Peter	21-Aug-1984	32.08835
6	Kate	30-Mar-1990	26.4853

Like in the previous example, the result of the YEARFRAC function is also a decimal number. To fix this, use the [ROUNDDOWN](#) function with 0 in the last argument because you don't want any decimal places.

So, here's an improved YEARFRAC formula to calculate age in Excel:

```
=ROUNDDOWN(YEARFRAC(B2, TODAY(), 1), 0)
```

```
=ROUNDDOWN(YEARFRAC(B2, TODAY(), 1), 0)
```

	A	B	C	D
1	<b>Name</b>	<b>Date of birth</b>	<b>Age</b>	
2	Mike	6-Jun-1996	20	
3	Natalie	20-Mar-1983	33	
4	Neal	16-Nov-2000	15	
5	Peter	21-Aug-1984	32	
6	Kate	30-Mar-1990	26	

## Calculate age in Excel with DATEDIF

One more way to convert date of birth to age in Excel is using the [DATEDIF](#) function:

```
DATEDIF(start_date, end_date, unit)
```

This function can return the difference between two dates in various time units such as years, months and days, depending on the value you supply in the **unit** argument:

Y - returns the number of **complete years** between the start and end dates.

M - returns the number of **complete months** between the dates.

D - returns the number of **days** between the two dates.

YM - returns **months**, ignoring days and years.

MD - returns the difference in **days**, ignoring months and years.

YD - returns the difference in **days**, ignoring years.

Since we aim to calculate **age in years**, we are using the "y" unit:

```
DATEDIF(date of birth, TODAY(), "y")
```

In this example, the DOB is in cell B2, and you reference this cell in your age formula:

```
=DATEDIF(B2, TODAY(), "y")
```

	A	B	C
1	Name	Date of birth	Age
2	Mike	6-Jun-1996	20
3	Natalie	20-Mar-1983	33
4	Neal	16-Nov-2000	15
5	Peter	21-Aug-1984	32
6	Kate	30-Mar-1990	26

No additional rounding function is needed in this case because a DATEDIF formula with the "y" unit calculates the number of full years:

	A	B	C
1	Name	Date of birth	Age
2	Mike	6-Jun-1996	20
3	Natalie	20-Mar-1983	33
4	Neal	16-Nov-2000	15
5	Peter	21-Aug-1984	32
6	Kate	30-Mar-1990	26

## How to get age from birthday in years, months and days

As you have just seen, calculating age as the number of full years that the person has lived is easy, but it is not always sufficient. If you want to know the exact age, i.e. how many years, months and days there are between someone's birth date and the current date, write 3 different DATEDIF functions:

1. To get the number of years: `=DATEDIF(B2, TODAY(), "Y")`
2. To get the number of months: `=DATEDIF(B2, TODAY(), "YM")`
3. To get the number of days: `=DATEDIF(B2, TODAY(), "MD")`

Where B2 is the date of birth.

And then, concatenate the above functions in a single formula, like this:

```
=DATEDIF(B2, TODAY(), "Y") & DATEDIF(B2, TODAY(), "YM") & DATEDIF(B2, TODAY(), "MD")
```

The above formula returns 3 numbers (years, months, and days) concatenated in a single text string, as shown in the screenshot below:

=DATEDIF(B2,TODAY(),"Y") & DATEDIF(B2,TODAY(),"YM") & DATEDIF(B2,TODAY(),"MD")								
	A	B	C	D	E	F	G	H
1	Name	Date of birth	Age					
2	Mike	6-Jun-1996	20320					
3	Natalie	20-Mar-1983	3366					
4	Neal	25-Sep-2000	1601					
5	Peter	21-Aug-1984	3215					
6	Kate	30-Mar-1990	26527					

Does not make much sense, uh? To make the results more meaningful, separate the numbers with commas and define what each value means:

```
=DATEDIF(B2,TODAY(),"Y") & " Years, " & DATEDIF(B2,TODAY(),"YM") & " Months, " &
DATEDIF(B2,TODAY(),"MD") & " Days"
```

The result looks much better now:

C2 : =DATEDIF(B2,TODAY(),"Y") & " Years, "& DATEDIF(B2,TODAY(),"YM") & " Months, "& DATEDIF(B2,TODAY(),"MD") & " Days"								
	A	B	C	D	E	F	G	H
1	Name	Date of birth	Age					
2	Mike	6-Jun-1996	20 Years, 3 Months, 20 Days					
3	Natalie	20-Mar-1983	33 Years, 6 Months, 6 Days					
4	Neal	25-Sep-2000	16 Years, 0 Months, 1 Days					
5	Peter	21-Aug-1984	32 Years, 1 Months, 5 Days					
6	Kate	30-Mar-1990	26 Years, 5 Months, 27 Days					

The formula works great, but you can improve it even further by hiding zero values. For this, add 3 [IF statements](#) that check for 0's, one per each DATEDIF:

```
=IF(DATEDIF(B2,TODAY(),"Y")=0,"",DATEDIF(B2,TODAY(),"y")&" years, ")& IF(DATEDIF(B2,
TODAY(),"ym")=0,"",DATEDIF(B2,TODAY(),"ym")&" months, ")& IF(DATEDIF(B2,
TODAY(),"md")=0,"",DATEDIF(B2,TODAY(),"md")&" days")
```

The following screenshot demonstrates the final Excel age formula in action - it returns age in years, months, and days, displaying only **non-zero** values:

```
=IF(DATEDIF(B2, TODAY(),"y")=0,"",DATEDIF(B2, TODAY(),"y")&" years, ")&
IF(DATEDIF(B2, TODAY(),"ym")=0,"",DATEDIF(B2, TODAY(),"ym")&" months, ")&
IF(DATEDIF(B2, TODAY(),"md")=0,"",DATEDIF(B2, TODAY(),"md")&" days")
```

	A	B	C	D	E
1	Name	Date of birth	Age		
2	Mike	6-Jun-1996	20 years, 3 months, 20 days		
3	Natalie	20-Mar-1983	33 years, 6 months, 6 days		
4	Neal	25-Sep-2000	16 Years, 1 Days		
5	Peter	21-Aug-1984	32 years, 1 months, 5 days		
6	Kate	30-Mar-1990	26 years, 5 months, 27 days		

**Tip.** If you are looking for an Excel formula to calculate age in **years and months**, take the above formula and remove the last IF(DATEDIF()) block that computes days.

## Specific formulas to calculate age in Excel

The generic age calculation formulas discussed above work great in most cases. In some situations, however, you may need something very specific. Of course, it's not possible to cover every and each scenario, but the following examples will give you some ideas on how you can tweak an age formula depending on your particular task.

### How to calculate age on a specific date in Excel

If you want to know someone's age at a certain date, use the [DATEDIF age formula](#) discussed above, but replace the TODAY() function in the 2<sup>nd</sup> argument with the specific date.

Assuming the birth date is in B1, the following formula will return age as of 1 January 2020:

```
=DATEDIF(B1, "1/1/2020","Y") & " Years, " & DATEDIF(B1, "1/1/2020","YM") & " Months, " &
DATEDIF(B1, "1/1/2020", "MD") & " Days"
```

To make your age formula more flexible, you can input the date in some cell and reference that cell in your formula:

```
=DATEDIF(B1, B2,"Y") & " Years, " & DATEDIF(B1,B2,"YM") & " Months, " & DATEDIF(B1,B2,
"MD") & " Days"
```

Where B1 is the DOB, and B2 is the date on which you want to calculate age.

B3		=DATEDIF(B1, B2, "Y") & " Years, "& DATEDIF(B1,B2,"YM") & " Months, "& DATEDIF(B1,B2, "MD") & " Days"
	A	B
1	Date of birth:	12-Apr-1988
2	Calculate at the date of:	01-Jan-2020
3	Age:	31 Years, 8 Months, 20 Days

## Calculate age in a certain year

This formula comes in handy in situations when the complete date to calculate at is not defined, and you know only the year.

Let's say you are working with a medical database, and your goal is to find out the patients' age at the time they underwent the last full medical examination.

Assuming the dates of birth are in column B beginning with row 3, and the year of the last medical examination is in column C, the age calculation formula goes as follows:

```
=DATEDIF(B3,DATE(C3, 1, 1),"y")
```

Because the exact date of the medical examination is not defined, you use the DATE function with an arbitrary date and month argument, e.g. DATE(C3, 1, 1).

The DATE function extracts the year from cell B3, makes a complete date using the month and day numbers that you supplied (1-Jan in this example), and passes that date to DATEDIF. As the result, you get the patient's age as of January 1 of a particular year:

D3		=DATEDIF(B3,DATE(C3, 1, 1),"y")		
	A	B	C	D
1	Name	Date of Birth	Last Medical Examination	
2			Year	Age
3	Mike	6-Jun-1996	2015	18
4	Natalie	20-Mar-1983	2016	32
5	Neal	25-Sep-2000	2014	13
6	Peter	21-Aug-1984	2012	27
7	Kate	30-Mar-1990	2013	22

## Find out a date when a person attains N years of age

Supposing your friend was born on 8 March 1978. How do you know on what date he completes his 50 years of age? Usually, you'd simply add 50 years to the person's birthdate. In Excel, you do the same using the DATE

function:

```
=DATE(YEAR(B2) + 50, MONTH(B2), DAY(B2))
```

Where B2 is the date of birth.

Instead of hard-coding the number of years in the formula, you can reference a certain cell where your users can input any number of years (F1 in the screenshot below):

=DATE(YEAR(B2) + \$F\$1, MONTH(B2), DAY(B2))						
	A	B	C	D	E	F
1	Name	Date of Birth	Attains N years		N years:	50
2	Mike	6-Jun-1996	6-Jun-2046			
3	Natalie	20-Mar-1983	20-Mar-2033			
4	Neal	25-Sep-2000	25-Sep-2050			
5	Peter	21-Aug-1984	21-Aug-2034			
6	Kate	30-Mar-1990	30-Mar-2040			

## Calculate age from day, month and year in different cells

When a birthdate is split into 3 different cells (e.g. year is in B3, month in C3 and day in D3), you can calculate age in this way:

Get the date of birth by using the DATE and DATEVALUE functions:

```
DATE(B3, MONTH(DATEVALUE(C3&"1")), D3)
```

Embed the above formula into DATEDIF to calculate age from date of birth in years, months, and days:

```
=DATEDIF(DATE(B3, MONTH(DATEVALUE(C3&"1")), D3), TODAY(), "y") & " Years, "&
DATEDIF(DATE(B3, MONTH(DATEVALUE(C3&"1")), D3), TODAY(), "ym") & " Months, "&
DATEDIF(DATE(B3, MONTH(DATEVALUE(C3&"1")), D3), TODAY(), "md") & " Days"
```

=DATEDIF(DATE(B3,MONTH(DATEVALUE(C3&"1")),D3),TODAY(),"y") & " Years, "& DATEDIF(DATE(B3,MONTH(DATEVALUE(C3&"1")),D3),TODAY(),"ym") & " Months, "& DATEDIF(DATE(B3,MONTH(DATEVALUE(C3&"1")),D3),TODAY(),"md") & " Days"						
	A	B	C	D	E	F
1	Name	Date of birth			Age	
2		Year	Month	Day		
3	Mike	1996	June	6	20 Years, 3 Months, 21 Days	
4	Natalie	1983	March	20	33 Years, 6 Months, 7 Days	
5	Neal	2000	September	25	16 Years, 0 Months, 2 Days	
6	Peter	1984	August	21	32 Years, 1 Months, 6 Days	
7	Kate	1990	May	30	26 Years, 3 Months, 28 Days	

	A	B	C
1	<b>Age Calculator for Excel</b>		
2			
3	<b>Date of birth</b>	15-Apr-2000	
4			
5	<b>Age at the date of:</b>		
6	Today's date	<input type="radio"/>	
7	Specific date	<input checked="" type="radio"/>	1-Jan-2020
8			
9	<b>Your age is</b>	19 Years, 8 Months, 17 Days	
10	<b>Your age in months</b>	236	
11	<b>Your age in days</b>	7200	

For this, add a couple of option buttons (*Developer tab > Insert > Form controls > Option Button*), and link them to some cell. And then, write an IF/DATEDIF formula to get age either at today's date or at the date specified by the user.

The formula works with the following logic:

If the *Today's date* option box is selected, value 1 appears in the linked cell (I5 in this example), and the age formula calculates based on the today date:

```
IF($I$5=1, DATEDIF($B$3,TODAY(),"Y") & " Years, " & DATEDIF($B$3,TODAY(),"YM") & " Months, " & DATEDIF($B$3, TODAY(), "MD") & " Days")
```

If the *Specific date* option button is selected AND a date is entered in cell B7, age is calculated at the specified date:

```
IF(ISNUMBER($B$7), DATEDIF($B$3, $B$7,"Y") & " Years, " & DATEDIF($B$3, $B$7,"YM") & " Months, " & DATEDIF($B$3, $B$7,"MD") & " Days", "")
```

Finally, nest the above functions into each other, and you will get the complete age calculation formula (in B9):  
`=IF($I$5=1, DATEDIF($B$3, TODAY(), "Y") & " Years, " & DATEDIF($B$3, TODAY(), "YM") & " Months, " & DATEDIF($B$3, TODAY(), "MD") & " Days", IF(ISNUMBER($B$7), DATEDIF($B$3, $B$7,"Y") & " Years, " & DATEDIF($B$3, $B$7,"YM") & " Months, " & DATEDIF($B$3, $B$7,"MD") & " Days", ""))`

The formulas in B10 and B11 work with the same logic. Of course, they are much simpler because they include just one DATEDIF function to return age as the number of complete months or days, respectively.


To learn the details, I invite you to download this Excel Age Calculator and investigate the formulas in cells B9:B11.

[Download Age Calculator for Excel](#)

## Ready-to-use age calculator for Excel

They users of our [Ultimate Suite for Excel](#) don't have to bother about making their own age calculator in Excel - it is only a couple of clicks away:

1. Select a cell where you want to insert an age formula, go to the *Ablebits Tools* tab > *Date & Time* group, and click the **Date & Time Wizard** button.



2. The Date & Time Wizard will start, and you go directly to the **Age** tab.
3. On the *Age* tab, there are 3 things for you to specify:
  - Data of birth* as a cell reference or a date in the mm/dd/yyyy format.
  - Age at today's date* or *specific date*.
  - Choose whether to calculate age in days, months, years, or exact age.
4. Click the **Insert formula** button.

Done!

	A	B	C	D	E	F	G
1	Date of birth	Age					
2	3-Mar-2011	=IF(AND(A2>TODAY(),OR(DATEDIF(IF(A2<TODAY(),A2,TODAY()),IF					
3	13-Apr-2003						
4	12-Jan-1989						
5	9-Dec-1967						
6							
7							
8							
9							
10							
11							
12							
13							
14							
15							
16							
17							
18							
19							
20							
21							
22							
23							
24							
25							
26							
27							
28							

**Date & Time Formula Wizard**

Add Subtract Difference **Age**

Arguments

Date of birth:  
A2

Age at the date of:  
 Today's date  
 Specific date:

Age:  
Y + M + D = 6 years 9 months 10 days  
Y = 6 years  
M = 81 months  
W = 353 weeks  
D = 2477 days  
Y + M + D = 6 years 9 months 10 days

Ablebits

The formula is inserted in the selected cell momentarily, and you double-click the fill handle to copy it down the column.

	A	B	C	D	E	F
1	Date of birth	Age				
2	3-Mar-2011	6 years 9 months 10 days				
3	13-Apr-2003	14 years 8 months 0 days				
4	12-Jan-1989	28 years 11 months 1 day				
5	9-Dec-1967	50 years 0 months 4 days				

As you may have noticed, the formula created by our Excel age calculator is more complex than the ones we've discussed so far, but it caters for singular and plural of time units such as "day" and "days".

If you'd like to get rid of zero units like "0 days", select the **Do not show zero units** check box:

`=IF(AND(A2>TODAY()),OR(DATEDIF(IF(A2<TODAY()),A2,TODAY()),IF(A`

	A	B	C	D
1	Date of birth	Age		
2	3-Mar-2011	6 years 9 months 10 days		
3	13-Apr-2003	14 years 8 months		
4	12-Jan-1989	28 years 11 months 1 day		
5	9-Dec-1967	50 years 4 days		

**Date & Time Formula Wizard**

Add Subtract Difference **Age**

Arguments

Date of birth:  
A2

Age at the date of:  
 Today's date  
 Specific date:

Age:  
Y + M + D = 6 years 9 months 10 days

Add text labels  
 Do not show zero units

Ablebits

If you are curious to test this age calculator as well as to discover 60 more time-saving add-ins for Excel, you are welcome to [download a trial version](#) of our Ultimate Suite. If you like the tools and decide to get a license, be sure to grab a 15% off coupon code that we provide to our valuable blog readers: AB14-BlogSpo

## How to highlight certain ages (under or over a specific age)

In some situations, you may need not only calculate age in Excel, but also highlight cells which contain ages that are under or over a particular age.


If your age calculation formula returns the number of complete years, then you can create a regular [conditional formatting rule](#) based on a simple formula like these ones:

To highlight ages equal to or greater than 18: `=<math>\$C2 \ge 18</math>`

To highlight ages under 18: `=<math>\$C2 < 18</math>`

Where C2 is the top-most cell in the Age column (not including the column header).

	A	B	C	D	E	F	G	H	I
1	<b>Name</b>	<b>Date of birth</b>	<b>Age</b>						
2	Mike	6-Jun-1996	20						
3	Natalie	20-Mar-1983	33						
4	Neal	16-Nov-2000	15						
5	Peter	21-Aug-1984	32						
6	Kate	30-Mar-1990	26						
7	Sean	15-Jan-2003	13						



But what if your formula displays age in years and months, or in years, months and days? In this case, you will have to create a rule based on a DATEDIF formula that [calculates age from date of birth in years](#).

Supposing the birthdates are in column B beginning with row 2, the formulas are as follows:

To highlight ages **under** 18 (yellow):

```
=DATEDIF($B2, TODAY(), "Y") < 18
```

To highlight ages **between** 18 and 65 (green):

```
=AND(DATEDIF($B2, TODAY(), "Y") >= 18, DATEDIF($B2, TODAY(), "Y") <= 65)
```


To highlight ages **over** 65 (blue):

```
=DATEDIF($B2, TODAY(), "Y") > 65
```

To create rules based on the above formulas, select the cells or entire rows that you want to highlight, go to the *Home* tab > *Styles* group, and click *Conditional Formatting* > *New Rule...* > *Use a formula to determine which cells to format*.

The detailed steps can be found here: [How to make a conditional formatting rule based on formula.](#)

	A	B	C	D	E	F	G
1	<b>Name</b>	<b>Date of birth</b>	<b>Age</b>				
2	Mike	6-Jun-1996	20 years, 3 months, 21 days				
3	Natalie	20-Mar-1983	33 years, 6 months, 7 days				
4	Neal	16-Nov-2000	15 years, 10 months, 11 days				
5	Peter	21-Aug-1942	74 years, 1 months, 6 days				
6	Kate	30-Mar-1990	26 years, 5 months, 28 days				
7	Sean	15-Jan-2003	13 years, 8 months, 12 days				



This is how you calculate age in Excel. I hope the formulas were easy for you to learn and you will give them a try in your worksheets. Thank you for reading and hope to see you on our blog next week!

## Practice workbook for download

[Excel Age Calculation examples \(.xlsx file\)](#)

## You may also be interested in

[How to calculate days between dates in Excel](#)

[How to convert date to year in Excel](#)

[Excel DATEDIF - calculate difference between two dates](#)

Copyright © 2003 - 2020 4Bits Ltd. All rights reserved. [Privacy policy](#) [Terms of use](#) [Contact us](#)

Microsoft and the Office logos are trademarks or registered trademarks of Microsoft Corporation. Google Chrome is a trademark of Google LLC.




# Age calculator in Excel

If you'd like to have your own age calculator in Excel, you can make one by using a few different DATEDIF formulas explained below. If you'd rather not reinvent the wheel, you can use the [age calculator](#) created by our Excel professionals.

## How to create an age calculator in Excel

Now that you know how to make an age formula in Excel, you can build a custom age calculator, for example this one:



Age Calculator for Excel	
Date of birth	<input type="text" value="17-Sep-1995"/>

What you see above is an embedded Excel Online sheet, so feel free to enter your birthdate in the corresponding cell, and you will get your age in a moment.

The calculator uses the following formulas to compute age based on the date of birth in cell A3 and today's date.

Formula in B5 calculates age in years, months, and days:

```
=DATEDIF(B2,TODAY(),"Y") & " Years, " & DATEDIF(B2,TODAY(),"YM") & " Months, " & DATEDIF(B2,TODAY(),"MD") & " Days"
```

Formula in B6 calculates age in months:

```
=DATEDIF($B$3,TODAY(),"m")
```

Formula in B7 calculates age in days:

```
=DATEDIF($B$3,TODAY(),"d")
```

If you have some experience with Excel Form controls, you can add an option to compute age at a specific date, like shown in the following screenshot: